/* Copyright 2009 the original author or authors.
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *      http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */
package geb

import static org.codehaus.groovy.runtime.StackTraceUtils.deepSanitize

class AtVerificationResult {
    private final AssertionError errorThrown
    private final boolean atResult

    AtVerificationResult(boolean atResult, AssertionError errorThrown) {
        this.atResult = atResult
        this.errorThrown = errorThrown
    }

    /**
     * Returns true if errorThrown is null and at verification result is true
     *
     * @return boolean value representing  at check verification result
     */
    boolean asBoolean() {
        errorThrown == null && atResult
    }

    String toString() {
        if (errorThrown) {
            def writer = new StringWriter()
            deepSanitize(errorThrown).printStackTrace(new PrintWriter(writer, true))
            writer
        } else {
            atResult
        }
    }

    void rethrowAnyErrors() {
        if (errorThrown) {
            throw errorThrown
        }
    }
}
